NOAA Eastern Region Computer Programs
and Problems NWS ERCP - No. 40

AVGPLOT and AVGCLIM

Alan Blackburn
National Weather Service Forecast Office
Buffalo, New York

Scientific Services Division
Eastern Region Headquarters
March 1987

U.S. DEPARTMENT OF COMMERCE / National Oceanic and Atmospheric Administration / National Weather Service

NOAA TECHNICAL MEMORANDUM

National Weather Service, Eastern Region Computer Programs and Problems

The Eastern Region Computer Programs and Problems (ERCP) series is a subset of the Eastern Region Technical Memorandum series. It will serve as the vehicle for the transfer of information about fully documented AFOS application programs. The format ERCP - No. 1 will serve as the model for future issuances in this series.

1 An AFOS version of the Flash Flood Checklist. Cynthia M. Scott, March 1981. (PB81 211252).

2 An AFOS Applications Program to Compute Three-Hourly Stream Stages. Alan P. Blackburn, September 1981. (PB82 156886).

3 PUPPY (AFOS Hydrologic Data Reporting Program). Daniel P. Provost, December 1981. (PB82 199720).

4 Special Search Computer Program. Alan P. Blackburn, April 1982. (PB83 175455).

5 Conversion of ALEMBIC$ Workbins. Alan P. Blackburn, October 1982. (PB83 138313).

6 Real-Time Quality Control of SAOs. John A. Billet, January 1983. (PB83 166082).

7 Automated Hourly Weather Collective from HRR Data Input. Lawrence Cedrone, January 1983 (PB83 167122).

8 Decoders for FRH, FTJ and FD Products. Cynthia M. Scott, February 1983. (PB83 176057).

9 Stability Analysis Program. Hugh M. Stone, March 1983. (PB83 197947).

10 Help for AFOS Message Comp. Alan P. Blackburn, May 1983. (PB83 213561).

11 Stability and Other Parameters from the First Transmission RAOB Data. Charles D. Little, May 1983. (PB83 220475).

12 TERR, PERR, and BIGC: Three Programs to Compute Verification Statistics. Matthew R. Peroutka, August 1983. (PB84 127521).

13 Decoder for Manually Digitized Radar Observations. Matthew R. Peroutka, June 1983. (PB84 127539).

14 Slick and Quick Data Entry for AFOS Era Verification (AEV) Program. Alan P. Blackburn, December 1983. (PB84 138726).

15 MDR--Processing Manually Digitized Radar Observations. Matthew R. Peroutka, November 1983. (PB84 161462) (Revised June 1985, PB85-220580/AS

16 RAWP: Stability Analysis Program. Hugh M. Stone, February 1984.(PB84 16144

17 ZONES. Gerald G. Rigdon, March 1984. (PB84 174325)

18 Automated Analysis of Upper Air Soundings to Specify Precipitation Type. Joseph R. Bocchieri and Gerald G. Rigdon, March 1984. (PB84 174333)

(Continued on Inside Rear Cover)

NOAA EASTERN REGION COMPUTER PROGRAMS AND PROBLEMS NWS ERCP - No. 40

AVGPLOT and AVGCLIM

Alan Blackburn
National Weather Service Forecast Office
Buffalo, New York

Scientific Services Division
Eastern Region Headquarters
March 1987

AVGPLOT and AVGCLIM

Alan Blackburn
National Weather Service Forecast Office
Buffalo, New York

I. INTRODUCTION

AVGPLOT and AVGCLIM manipulate monthly data by computing and plotting running averages, grouping them by months, and correlating above and below average months.

These programs were written to recreate a study of conditional probabilities by Bernard (1) at Buffalo, New York. The several moves and instrument changes at Buffalo required finding which moves affected normals, and finding a better way to determine normals and which years were above and below normal.

AVGPLOT plots running averages on an AFOS-displayable graphic. The plots of different lengths of averages show where changes had an effect on climatic data and where there were true climatic trends. You can specify station moves in AVGPLOT to eliminate contamination of averages between changes. AVGPLOT can also be used to check the input data by using a running average of 1 year.

AVGCLIM compares different groups of monthly data to show correlations between above and below normal. For example you can see how often a below normal February follows a below normal January. The groups are stratified into much below, below, normal, above and much above. Either one or two preceding groups can be correlated with another group. Separate input data files allow correlation between different types of data such as between temperature and snowfall.

II. METHODOLOGY AND SOFTWARE STRUCTURE

A. Description

Both AVGPLOT and AVGCLIM are run from the dasher where you type in the running average period, any station moves, and the grouping of months. Running averages are computed for each group as a whole and for each year, except near the beginning or end of a period. For example, the last 15 years would be the same if the running average were 30 years. AVGPLOT creates a graph of the running average which varies with the range of data.

AVGCLIM uses the running averages and mean to compute the standard deviation. .67 and 1.15 standard deviations are then used to stratify the data. Normal distribution tables show that this should give 50% in the normal category and 12.5% in each of the other four categories. The data are then placed into a contingency table by category.

B. Input and Output Files

AVGPLOT uses MONTHDAT1.DT for input. AVGCLIM uses MONTHDAT1.DT and MONTHDAT3.DT. MONTHDAT2.DT is used with a second preceding monthly group. The data begins with the year, followed with monthly values in three digits. Negative signs and values over 100 are placed in the 4th digit preceding.

Example: 1901 266 178 326 458 542 660 738 715 642 528 362 266 JAN=26.6
1902 095-012 392 450 547 613 698 670 642 515 478 276 FEB=-1.2

AVGPLOT output is in RDOS file AVGPRINT.DT in NMCGPHTØ8. AVGCLIM output is in AVGPRINT.DT. See figures for examples.

III. CAUTIONS AND RESTRICTIONS

When grouping months, enter them in chronological order i.e., use 12 then 1 and 2 for a winter group. If two groups have the same months, you can specify that they are from the same year if you want to compare different types of data.

IV. REFERENCES

1. Bernard, Harold, 1984: Conditional Probabilities of Winter Season Temperature Categories for Boston Massachusetts. National Weather Digest, Vol. 9, no. 1, pp 5-8, February 1984.

AVGPLOT

PART A: INFORMATION AND INSTALLATION

PROGRAM NAME: AVGPLOT

AAL ID:
REVISION NO.: 1.10

PURPOSE: Plot running averages of monthly data.

PROGRAM INFORMATION:

Development Programmer:
Alan Blackburn
Location: WSFO Buffalo, NY
Phone: (FTS) 437-4800

Maintenance Programmer:
Same
Location: WSFO Buffalo, NY
Phone: (FTS) 437-4800

Language: FORTRAN IV/Rev. 5.57
Date: 4/3/87
Running Time: 20 seconds for 112 years
Disk Space:
Program files 48 RDOS blocks
Data files 5 + 1 RDOS block per 10 years

PROGRAM REQUIREMENTS

Program Files:
AVGPLOT.SV

Data Files:

NAME	DP Location	R/W	Comment
MONTHDAT1.DT	SYSZ	R	Input Data
AVGPRINT.DT	SYSZ	W	list of averages

AFOS Products:

ID	Action
NMCGPHTØ8	Output

LOAD LINE

```
RLDR AVGPLOT AVGPLOTREV AG.LB BG.LB UTIL.LB FORT.LB
```

PROGRAM INSTALLATION

1. Create MONTHDAT1.DT with M:F/MONTHDAT1.DT or text editor.

   Sample Line: 1980 367 305 311 430 588 658 691 680 623 486 319 219

2. Put AVGPLOT.SV in APPL1, link to SYSZ.

AVGPLOT

PART B: PROGRAM EXECUTION and ERROR CONDITIONS

PROGRAM NAME: AVGPLOT

PROGRAM EXECUTION

1. Run from the dasher. Enter running average period, station moves and monthly groupings when requested.

ERROR CONDITIONS

None

AVGCLIM

PART A: INFORMATION AND INSTALLATION

PROGRAM NAME: AVGCLIM

AAL ID:
REVISION NO.: 1.00

PURPOSE: Compute and correlate averages of groups of months.

PROGRAM INFORMATION:

Development Programmer:
Alan Blackburn
Location: WSFO Buffalo, NY
Phone: (FTS) 437-4800

Maintenance Programmer:
Same
Location: WSFO Buffalo, NY
Phone: (FTS) 437-4800

Language: FORTRAN IV/Rev. 5.57
Date: 2/6/87
Running Time: 18 seconds for 112 years
Disk Space:
Program files 35 RDOS blocks
Data files 46 RDOS blocks

PROGRAM REQUIREMENTS

Program Files:
AVGCLIM.SV

Data Files:

Name	DP Location	R/W	Comments
MONTHDAT1.DT	SYSZ	R	Input - preceding group of months
MONTHDAT2.DT	SYSZ	R	Input - 2nd preceding group of months (optional)
MONTHDAT3.DT	SYSZ	R	Input - succeeding month(s)
AVGPRINT.DT	SYSZ	W	

AFOS Products:
NONE

LOAD LINE

```
RLDR AVGCLIM AVGCLIMREV BG.LB UTIL.LB FORT.LB
```

PROGRAM INSTALLATION

1. Create MONTHDAT1.DT with M:F/MONTHDAT1.DT or text editor.
   Sample Line: 1980 367 305 311 430 588 658 691 680 623 486 319 219
2. Create MONTHDATA2.DT and MONTHDAT3.DT with XFER or MOVE commands or use M:F/ if data is different.
3. Put AVGCLIM.SV in APPL1 and link to SYSZ.

AVGCLIM

PART B: PROGRAM EXECUTION and ERROR CONDITIONS

PROGRAM NAME: AVGCLIM

PROGRAM EXECUTION

1. Run from the dasher. Enter running average period, station moves and monthly groupings when requested.

ERROR CONDITIONS

None

VI. Figures

```
AVGPLOT
ENTER RUNNING AVERAGE PERIOD 30
WAS THE STATION MOVED? 1=YES OR 2=NO 1
WHICH YEAR? 1944
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 1961
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 99
LIST MONTH(S) TO BE AVERAGED
IF ENTIRE YEAR THEN TYPE 13 2
NEXT MONTH? IF NONE TYPE 99 3
NEXT MONTH? IF NONE TYPE 99 99
OUTPUT IN AVGPRINT.DT AND AVGRAPH
R
```

Figure 1

Sample dasher run for AVGPLOT

(This run will compute 30 year running averages for February and March data combined.)

YEAR	AVG	RAVG
1872	26.8	27.6
1873	28.5	27.6
1874	27.9	27.6
1875	20.2	27.6
1876	28.0	27.6
1877	29.2	27.6
1878	33.9	27.6
1879	25.5	27.6
1880	30.8	27.6
1881	25.6	27.6
1882	33.2	27.6
1883	23.9	27.6
1884	28.5	27.6
1885	17.5	27.6
1886	28.2	27.6
1887	27.5	27.3
1888	24.9	27.4
1889	26.0	27.6
1890	30.3	27.5
1891	30.8	27.7
1892	28.3	27.6
1893	25.9	27.5
1894	30.8	27.3
1895	21.7	27.5
1896	25.1	27.5
1897	31.0	27.6
1898	34.2	27.3
1899	26.5	27.4

Figure 2

AVGPRINT.DT Output from AVGPLOT



Figure 3
AVGRAPH Output from AVGPLOT

```
1871 272 283 409 482 568 660 689 707 582 523 347 269
1872 250 266 269 457 540 663 733 733 647 501 370 254
1873 249 257 313 417 548 678 693 672 592 486 310 319
1874 284 252 306 346 520 638 686 667 658 501 395 290
1875 178 134 270 374 518 636 679 670 582 476 340 308
1876 324 270 290 394 510 676 699 713 586 460 400 195
1877 201 304 280 435 525 642 694 707 640 521 390 367
1878 269 284 393 508 532 628 730 706 633 524 390 258
1879 206 193 317 390 548 623 706 671 592 582 380 304
1880 367 305 311 430 588 658 691 680 623 486 319 219
1881 178 216 295 362 566 592 698 717 712 540 411 368
1882 260 326 338 403 490 622 672 682 622 553 380 278
1883 211 235 242 407 502 646 668 658 578 484 434 334
1884 179 274 296 403 528 667 652 680 655 520 382 298
1885 210 146 204 407 538 608 700 650 602 500 420 310
1886 228 243 320 473 536 663 679 672 632 524 382 246
1887 228 265 284 409 622 662 750 680 590 478 385 309
1888 182 239 258 390 538 647 679 678 590 451 406 308
1889 298 185 334 434 546 619 699 674 626 452 406 375
1890 328 314 291 437 520 666 696 658 599 500 394 254
1891 278 305 310 436 514 655 653 678 652 520 394 366
1892 230 282 283 413 524 664 698 698 622 503 366 270
1893 168 212 305 410 527 674 700 688 602 535 402 276
1894 306 228 308 451 546 656 707 671 654 539 361 330
1895 229 165 269 448 562 688 680 691 648 456 402 332
1896 248 248 253 465 585 658 702 687 602 474 426 302
1897 249 276 344 442 538 616 740 670 632 544 402 303
1898 283 273 411 434 566 672 730 711 666 540 396 294
1899 254 212 318 474 562 664 700 718 602 558 420 311
1900 276 234 250 435 550 662 704 730 676 605 417 314
                      :
                      :
                      :
```

Figure 4

Sample MONTHDAT1.DT Input

```
AVGCLIM
ENTER RUNNING AVERAGE PERIOD 30
WAS THE STATION MOVED? 1=YES OR 2=NO 1
WHICH YEAR? 1944
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 1961
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 99
LIST FIRST GROUP OF PRECEDING MONTHS
IF ENTIRE YEAR THEN TYPE 13 12
NEXT MONTH? IF NONE TYPE 99 1
NEXT MONTH? IF NONE TYPE 99 2
NEXT MONTH? IF NONE TYPE 99 99
LIST SECOND GROUP OF PRECEDING MONTHS
TYPE 13 FOR ENTIRE YEAR, TYPE 99 IF NO SECOND GROUP 3
NEXT MONTH? IF NONE TYPE 99 4
NEXT MONTH? IF NONE TYPE 99 5
NEXT MONTH? IF NONE TYPE 99 99
LIST MONTHS TO BE FORECAST
IF ENTIRE YEAR THEN TYPE 13 6
NEXT MONTH? IF NONE TYPE 99 7
NEXT MONTH? IF NONE TYPE 99 8
NEXT MONTH? IF NONE TYPE 99 99
OUTPUT IN AVGPRINT.DT
R
```

Figure 5

Sample dasher run for AVGCLIM

(This run will determine the categorical distribution of summer data for each combination of winter and spring categories. See output in Figure 6.)

		FORECAST				
1ST	2ND	MB	B	N	A	MA
MB	MB	0	0	0	0	1
MB	B	0	1	1	1	0
MB	N	1	0	1	1	2
MB	A	0	0	4	0	0
MB	MA	1	0	1	0	1
B	MB	0	0	2	0	0
B	B	0	0	0	0	0
B	N	0	1	2	1	1
B	A	0	0	1	0	0
B	MA	0	0	1	0	0
N	MB	2	4	3	0	0
N	B	0	0	5	1	3
N	N	5	6	18	5	3
N	A	0	1	0	0	2
N	MA	2	1	2	0	2
A	MB	0	1	0	0	0
A	B	0	0	0	0	0
A	N	0	0	1	1	0
A	A	0	0	0	1	0
A	MA	0	0	1	0	1
MA	MB	2	1	0	0	0
MA	B	0	0	2	0	0
MA	N	0	2	4	2	0
MA	A	0	0	0	0	0
MA	MA	0	0	1	0	1

ABOVE INTERVAL 1ST: 2.1 2ND 1.5 FCST 1.0
MUCH ABOVE INTERVAL 1ST: 3.6 2ND 2.6 FCST 1.7

YEAR	AVG	RAVG	AVG	RAVG	AVG	RAVG
1872	26.2	26.1	42.6	42.4	68.1	67.8
1873	25.3	26.1	39.1	42.4	66.4	67.8
1874	28.5	26.1	38.7	42.4	66.2	67.8
1875	20.1	26.1	39.8	42.4	69.6	67.8
1876	30.1	26.1	41.3	42.4	68.1	67.8
1877	23.3	26.1	47.8	42.4	68.8	67.8
1878	30.7	26.1	41.8	42.4	66.7	67.8
1879	21.9	26.1	44.3	42.4	67.6	67.8
1880	32.5	26.1	40.8	42.4	66.9	67.8
1881	20.4	26.1	41.0	42.4	65.9	67.8
1882	31.8	26.1	38.4	42.4	65.7	67.8
1883	24.1	26.1	40.9	42.4	66.6	67.8
1884	26.2	26.1	38.3	42.4	65.3	67.8
1885	21.8	26.1	44.3	42.4	67.1	67.8
1886	26.0	26.1	43.8	42.2	69.7	67.8
1887	24.6	26.3	39.5	42.4	66.8	67.7
1888	24.3	26.2	43.8	42.6	66.4	67.6
1889	26.4	26.3	41.6	42.7	67.3	67.6
1890	33.9	26.0	42.0	42.8	66.2	67.6
1891	27.9	26.0	40.7	42.9	68.7	67.6
1892	29.3	26.0	41.4	42.8	68.7	67.5
1893	21.7	26.1	46.2	42.7	67.8	67.5
1894	27.0	25.9	42.6	42.7	68.6	67.6
1895	24.1	26.2	43.4	42.8	68.2	67.5
1896	27.6	25.9	44.1	42.9	67.5	67.6
1897	27.6	26.1	47.0	42.9	70.4	67.6

Figure 6

Sample output AVGPRINT.DT for AVGCLIM

```
AVGCLIM
ENTER RUNNING AVERAGE PERIOD 30
WAS THE STATION MOVED? 1=YES OR 2=NO 1
WHICH YEAR? 1944
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 1961
IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 99
LIST FIRST GROUP OF PRECEDING MONTHS
IF ENTIRE YEAR THEN TYPE 13 1
NEXT MONTH? IF NONE TYPE 99 99
LIST SECOND GROUP OF PRECEDING MONTHS
TYPE 13 FOR ENTIRE YEAR, TYPE 99 IF NO SECOND GROUP 99
LIST MONTHS TO BE FORECAST
IF ENTIRE YEAR THEN TYPE 13 2
NEXT MONTH? IF NONE TYPE 99 99
OUTPUT IN AVGPRINT.DT
0
```

Figure 7

Sample dasher run for AVGCLIM with only two periods

(This run will determine the categorical distribution of February data for each category of January data.)

		FORECAST				
		MB	B	NM	A	MA
1ST	MB	3	3	8	0	0
PER	B	2	1	6	1	3
	NORM	9	4	29	10	7
	A	2	0	4	4	2
	MA	0	2	6	2	4

ABOVE INTERVAL OBS: 3.2 FCST: 3.0
MUCH ABOVE INTERVAL OBS: 5.5 FCST: 5.1

YEAR	AVG	RAVG	AVG	RAVG
1872	25.0	24.7	26.6	24.7
1873	24.9	24.7	25.7	24.7
1874	28.4	24.7	25.2	24.7
1875	17.8	24.7	13.4	24.7
1876	32.4	24.7	27.0	24.7
1877	20.1	24.7	30.4	24.7
1878	26.9	24.7	28.4	24.7
1879	20.6	24.7	19.3	24.7
1880	36.7	24.7	30.5	24.7
1881	17.8	24.7	21.6	24.7
1882	26.0	24.7	32.6	24.7
1883	21.1	24.7	23.5	24.7
1884	17.9	24.7	27.4	24.7
1885	21.0	24.7	14.6	24.7
1886	22.8	24.7	24.3	24.7
1887	22.8	24.7	26.5	24.4
1888	18.2	24.7	23.9	24.2
1889	29.8	24.7	18.5	24.3
1890	32.8	24.4	31.4	24.0
1891	27.8	24.5	30.5	24.1
1892	23.0	24.5	28.2	24.1
1893	16.8	24.7	21.2	23.7
1894	30.6	24.7	22.8	23.5
1895	22.9	24.9	16.5	23.8
1896	24.8	24.6	24.8	23.5
1897	24.9	24.9	27.6	23.7
1898	28.3	24.6	27.3	23.2
1899	25.4	25.0	21.2	23.2
1900	27.6	25.3	23.4	22.8
1901	26.6	25.5	17.8	23.3
1902	24.8	25.8	22.2	23.2
1903	26.2	25.8	27.0	22.9
1904	19.2	25.7	16.8	22.9
1905	20.0	25.8	17.6	23.2
1906	33.4	25.2	25.2	22.8
1907	25.4	25.2	19.4	22.8
1908	26.3	25.2	21.4	22.7
1909	28.7	25.5	29.6	22.7
1910	26.2	25.3	21.9	22.7
1911	28.2	25.3	26.5	23.2
1912	15.6	25.3	19.6	23.1
1913	33.8	25.2	22.2	23.1
1914	27.9	25.2	16.9	23.1
1915	25.3	25.1	29.6	23.1
1916	32.0	25.0	18.9	23.3
1917	24.4	25.0	18.0	23.6
1918	14.1	25.4	23.1	23.9
1919	31.0	25.7	28.8	23.9

Figure 8

AVGPRINT.DT output from AVGCLIM with two periods.

VII. Source Code

```
C     PROGRAM AVGPLOT.                                        REV 01.10
C     DEC 13, 1984          BLACKBURN, ALAN P    WSFO BUF/FTS 437-4800
C     FORTRAN IV/ REV 5.20  DG ECLIPSE (S230)    RDOS/REV 6.18
C     LOAD LINE:  RLDR AVGPLOT AG.LB OUT.RB BG.LB UTIL.LB FORT.LB
C                 AFOSE.LB
C     PURPOSE
C         PLOTS AND PRINTS RUNNING AVERAGES AND AVERAGES OF MONTHLY
C         DATA ALLOWING FOR STATION AND INSTRUMENT CHANGES
C     EXTERNALS
C         UTIL.LB: GCHN OPENR RDL UNPACK DFILW CRAND KLOSE
C         FORT.LB: OPEN
C         AG.LB:   LINES TEXT UTF
C         OUT.RB
C     CHANNELS/FILES
C         IC - ASSIGNED TO INPUT FILE MONTHDAT1.DT
C         JC - ASSIGNED TO OUTPUT FILE AVGPRINT.DT
C         OUTPUT IN AVGRAPH THROUGH AG.LB
C     VARIABLES
C       NAV      LENGTH OF RUNNING AVERAGE      MIDAV   AVERAGING MIDPOINT
C       NPDS     NUMBER OF SEPARATE AVERAGED    MV()    DATES OF MOVES
C                  PERIODS                      MONTH() MONTHS TO BE AVERAGED
C       NMON1    # OF MONTHS TO BE AVERAGED     K1      MONTH WHICH COMES FIRST
C       ISTYR    STARTING YEAR                            IN THE YEAR
C       AVG()    AVERAGES                       J       TOTAL YEARS
C       PD()     LENGTH OF PERIODS              JYR     STARTING YEAR OF PERIOD
C       RAV()    RUNNING AVERAGES               IVL     X-AXIS INTERVAL
C       MN,RN    MINIMUM RUNNING AVERAGE        MX,RX   MAXIMUM RUNNING AVERAGE
C       VVL      VERTICAL MULTIPLIER            LX,LY   COORDINATES FOR PLOTTING
C       VAL, IVAL VALUES OF GRID LABELS
C
C
C
      DIMENSION IBUF(50),IU(60),LX(150),LY(150),MV(10),MONTH(3,12)
      INTEGER AVG(150),RAV(150),PD(10),SCRIPT(3)
C
C
      ACCEPT "ENTER RUNNING AVERAGE PERIOD ",NAV
      MIDAV=NAV/2+1                          ; GET MIDPOINT OF AVERAGING
      NPDS=1
      MV(1)=2000
      ACCEPT "WAS THE STATION MOVED? 1=YES OR 2=NO ",IANS
      IF(IANS.EQ.2)GOTO 20
      ACCEPT "WHICH YEAR? ",MV(NPDS)
10    NPDS=NPDS+1
      ACCEPT "IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 ",
     2IANS
      IF (IANS.EQ.99)GOTO 20
      MV(NPDS)=IANS
      GOTO 10
20    CONTINUE
C
C
      DO 22 I=1,12
22      MONTH(1,I)=I
C
      NMON1=1
      K1=1
      TYPE "LIST MONTH(S) TO BE AVERAGED"
      ACCEPT "IF ENTIRE YEAR THEN TYPE 13 ",MONTH(1,1)
      IF(MONTH(1,1).EQ.13)NMON1=12
```

```
      IF(MONTH(1,1).EQ.13)GOTO 28
      DO 25 I=2,12
        ACCEPT "NEXT MONTH? IF NONE TYPE 99 ",MONTH(1,I)
        IF(MONTH(1,I).EQ.99)GOTO 26
25      NMON1=NMON1+1                          ; NMON1 IS NUMBER OF MONTHS
26      IF(NMON1.EQ.1)GOTO 28
      DO 27 I=2,NMON1
27      IF(MONTH(1,I).LT.MONTH(1,(I-1)))K1=I   ; K1 IS FIRST MONTH OF YEAR
C
28    CALL GCHN(IC,IER)
      CALL OPENR(IC,"MONTHDAT1.DT",0,IER)
C
      IF(MONTH(1,1).EQ.13)MONTH(1,1)=1
      ISUM=0
      DO 100 J=1,150
        CALL RDL(IC,IBUF,N,IER)
        IF(IER.NE.1)GOTO 110
        CALL UNPACK(IBUF,N,IU)
C
        IF(J.NE.1)GOTO 40                      ; GET STARTING YEAR NEXT LINE
        ISTYR=(IU(1)-48)*1000+(IU(2)-48)*100+(IU(3)-48)*10+IU(4)-48
C
40      DO 50 I=K1,NMON1                       ; SUM MONTHS
          ISUM=(IU(2+MONTH(1,I)*4)-48)*100+ISUM ; CONVERT TO DECIMAL
          ISUM=(IU(3+MONTH(1,I)*4)-48)*10+ISUM
          ISUM=IU(4+MONTH(1,I)*4)-48+ISUM
          IF(IU(1+MONTH(1,I)*4).EQ.45)ISUM=ISUM*(-1) ; NEGATIVE VALUES
50        IF(IU(1+MONTH(1,I)*4).EQ.49)ISUM=ISUM+1000 ; OVER 100
        TSUM=(ISUM*1.0)/NMON1+0.5
        AVG(J)=TSUM
C
        ISUM=0
        IF(K1.EQ.1)GOTO 100                    ; MONTHS ARE ALL IN SAME YEAR
        KX=K1-1
        DO 55 I=1,KX                           ; GET SUM STARTED OVER NEW YEAR
          ISUM=(IU(2+MONTH(1,I)*4)-48)*100+ISUM
          ISUM=(IU(3+MONTH(1,I)*4)-48)*10+ISUM
55        ISUM=IU(4+MONTH(1,I)*4)-48+ISUM
C
100   CONTINUE
110   CONTINUE
      J=J-1                                    ; J IS TOTAL YEARS OF RECORD
C
C
      PD(1)=J
      IF(NPDS.EQ.1)GOTO 130
      PD(1)=MV(1)-ISTYR                        ; GET PERIODS OF RECORD
      MV(NPDS)=ISTYR+J
      DO 120 I=2,NPDS
120     PD(I)=MV(I)-MV(I-1)
C
C
130   JYR=0                                    ; JYR IS START OF PERIOD
      DO 190 I=1,NPDS
        IF(PD(I).LT.NAV)GOTO 170
C
C
      KST=JYR+MIDAV
      KEND=JYR+PD(I)-(NAV-MIDAV)
      DO 150 K=KST,KEND                ; CYCLE THROUGH MIDDLE OF PERIOD
```

```
        RAV(K)=0
        DO 140 K2=1,NAV
140     RAV(K)=RAV(K)+AVG(K-MIDAV+K2)        ; SUM AVERAGES
150     RAV(K)=RAV(K)/NAV                    ; RAV IS RUNNING AVERAGE
      IF(NAV.EQ.1)GOTO 185
C
C
      KEND=MIDAV-1
      DO 154 K=1,KEND
154     RAV(JYR+K)=RAV(JYR+MIDAV)            ; FILL IN BEGINNING OF PERIOD
      KEND=MIDAV
      DO 156 K=1,KEND
156     RAV(PD(I)-K+1+JYR)=RAV(JYR+PD(I)-MIDAV-1) ; FILL IN END
      GOTO 185
C
C
C.........PERIOD LESS THAN AVERAGING PERIOD
170   RAVG=0
      KST=JYR+1
      KEND=JYR+PD(I)
      DO 175 K=KST,KEND
175     RAVG=RAVG+AVG(K)
      RAVG=RAVG/PD(I)
      DO 180 K=KST,KEND
180     RAV(K)=RAVG
185   JYR=JYR+PD(I)
190   CONTINUE
C
C
C
C
C
C.........PLOTTING SECTION
      IVL=3500/J                           ; X DIRECTION INCREMENT
      RN=1200
      RX=-500
      DO 195 I=2,J                  ; GET MIN AND MAX RUNNING AVERAGE
        IF(RAV(I).LT.RN)RN=RAV(I)
195     IF(RAV(I).GT.RX)RX=RAV(I)
      MN=10*INT(RN/10.0)                   ; ROUND OFF MIN AND MAX
      MX=10*INT(RX/10.0+0.9)
      IDIFF=MX-MN
      VVL=125                              ; VVL IS VERTICAL MULTIPLIER
      IF(IDIFF.GT.20)VVL=50
      IF(IDIFF.GT.50)VVL=25
      IF(IDIFF.GT.100)VVL=12.5
      IF(IDIFF.GT.200)VVL=6.25
      DO 200 K=2,J
        LX(K-1)=100+IVL*K                  ; MOVE IN X DIRECTION
200     LY(K-1)=INT(250+VVL*(RAV(K)-MN))   ; MOVE IN Y DIRECTION
      CALL LINES(LX,LY,J-1,1,0)            ; DRAW LINE
C
C
C.........DRAW HORIZONTAL LINES
      LX(1)=100
      LX(2)=3500
      DO 300 K=1,11
        LY(1)=250+250*(K-1)
        LY(2)=LY(1)
300     CALL LINES(LX,LY,2,1,0)
```

```
C
C
C.............DRAW VERTICAL LINES
      J2=J/10+1
      LY(1)=250
      LY(2)=2750
      DO 320 K=1,J2
        LX(1)=100+(K-1)*IVL*10
        LX(2)=LX(1)
320     CALL LINES(LX,LY,2,1,0)
      CALL DFILW("AVGRAPH")
      CALL CRAND("AVGRAPH")
C
C
C
C
C...........LABELS
      SCRIPT(2)=0
      IF(VVL.EQ.125)GOTO 410
      DO 400 I=1,6
        VAL=(MN+(I-1)*500/VVL)/10          ; VALUE OF LABEL
        IVAL10=VAL/10.0                    ; PUT INTO ASCII
        IVAL=VAL-10*IVAL10+48
        IVAL10=IVAL10+48
        SCRIPT(1)=ISHFT(IVAL10,8)+IVAL
400     CALL TEXT(SCRIPT,25,(I*500-250),0,1,0,0)
      GOTO 430
C
C
410   DO 420 I=1,3
        VAL=(MN+(I-1)*10)/10
        IVAL10=VAL/10.0
        IVAL=VAL-10*IVAL10+48
        IVAL10=IVAL10+48
        SCRIPT(1)=ISHFT(IVAL10,8)+IVAL
420     CALL TEXT(SCRIPT,25,(250+(I-1)*1250),0,1,0,0)
430   CONTINUE
C
C
C.........HORIZONTAL LABELS
      IYR1=ISTYR-1                         ; IYR1 IS BEGINNING YEAR
      J2=J2/2                              ; EVERY 20 YEARS
      SCRIPT(3)=0
      DO 440 I=1,J2
        IYR=IYR1+(I-1)*20-1000             ; COMPUTE YEAR
        IVALC=IYR/100                      ; PUT INTO ASCII
        IYR=IYR-IVALC*100
        IVAL10=IYR/10
        IVAL=IYR-10*IVAL10+48
        SCRIPT(1)=ISHFT(49,8)+IVALC+48
        SCRIPT(2)=ISHFT((IVAL10+48),8)+IVAL
440     CALL TEXT(SCRIPT,(27+20*(I-1)*IVL),200,0,1,0,0)
      CALL UTF("NMCGPHT08","AVGRAPH")
C
C
C.........PRINTED OUTPUT
      CALL GCHN(JC,IER)
      CALL DFILW("AVGPRINT.DT")
      CALL CRAND("AVGPRINT.DT")
      CALL OPEN(JC,"AVGPRINT.DT",0,IER)
```

```
      WRITE(JC,905)
905   FORMAT('  YEAR  AVG RAVG')
      DO 330 I=2,J
        PAVG=AVG(I)*0.1
        PRAV=RAV(I)*0.1
        IYR=ISTYR+I-1
330     WRITE(JC,900) IYR,PAVG,PRAV
900   FORMAT(I6,2F5.1)
      TYPE 'OUTPUT IN AVGPRINT.DT AND NMCGPHT08'
      CALL KLOSE
      STOP
      END
```

```
C     PROGRAM AVGCLIM.                                    REV 01.00
C     DEC 14, 1984          BLACKBURN, ALAN P    WSFO BUF/FTS 437-4800
C     FORTRAN IV/ REV 5.20  DG ECLIPSE (S230)    RDOS/REV 6.18
C     LOAD LINE:  RLDR AVGPLOT BG.LB UTIL.LB FORT.LB AFOSE.LB
C     PURPOSE
C         COMPUTES AVERAGES OF GROUPS OF MONTHS AND CORRELATES THEM IN TERMS OF
C         BELOW OR ABOVE AVERAGE.
C     EXTERNALS
C         UTIL.LB: GCHN OPENR RDL UNPACK DFILW CRAND KLOSE
C         FORT.LB: OPEN
C     CHANNELS/FILES
C         IC - ASSIGNED TO INPUT FILES MONTHDAT1.DT MONTHDAT2.DT MONTHDAT3.DT
C         JC - ASSIGNED TO OUTPUT FILE AVGPRINT.DT
C     VARIABLES
C       NAV      LENGTH OF RUNNING AVERAGE      MIDAV   AVERAGING MIDPOINT
C       NPDS     NUMBER OF SEPARATE AVERAGED    MV()    DATES OF MOVES
C                  PERIODS                      MONTH() MONTHS TO BE AVERAGED
C       NMON()   # OF MONTHS TO BE AVERAGED     K1()    MONTH WHICH COMES FIRST
C       ISTYR    STARTING YEAR                            IN THE YEAR
C       AVG()    AVERAGES                       J       TOTAL YEARS
C       PD()     LENGTH OF PERIODS, LABELS      JYR     STARTING YEAR OF PERIOD
C       RAV()    RUNNING AVERAGES               ISAME() MONTHS IN SAME YEAR
C       SD()     STANDARD DEVIATION             ICAT()  CATEGORIES OF YEARS
C
C
C
C
      DIMENSION IBUF(50),IU(60),MV(10),MONTH(3,12),NMON(3),K1(3),SD(3)
      INTEGER AVG(3,150),RAV(3,150),PD(10),ICAT(3,150)
      DIMENSION MAT2(5,5),MAT3(5,5,5),PA(3),PS(3),ISAME(3)
C
C
      ACCEPT "ENTER RUNNING AVERAGE PERIOD ",NAV
      MIDAV=NAV/2+1                          ; GET MIDPOINT OF AVERAGING
      NPDS=1
      MV(1)=2000
      ACCEPT "WAS THE STATION MOVED? 1=YES OR 2=NO ",IANS
      IF(IANS.EQ.2)GOTO 20
      ACCEPT "WHICH YEAR? ",MV(NPDS)
10    NPDS=NPDS+1
      ACCEPT "IF THE STATION MOVED AGAIN TYPE YEAR, OTHERWISE TYPE 99 ",
     2IANS
      IF (IANS.EQ.99)GOTO 20
      MV(NPDS)=IANS
      GOTO 10
20    CONTINUE
C
C


      NMON(1)=1
      NMON(2)=1
      NMON(3)=1
      DO 115 L=1,3
      DO 22 I=1,12
22      MONTH(L,I)=I
C
C
      K1(L)=1
      IF(L.NE.1)GOTO 24
      TYPE "LIST FIRST GROUP OF PRECEDING MONTHS"
```

```
      ACCEPT "IF ENTIRE YEAR THEN TYPE 13 ",MONTH(1,1)
24    IF(L.NE.2)GOTO 25
      TYPE "LIST SECOND GROUP OF PRECEDING MONTHS"
      ACCEPT "TYPE 13 FOR ENTIRE YEAR, TYPE 99 IF NO SECOND GROUP ",
     2MONTH(2,1)
      IF(MONTH(2,1).EQ.99)GOTO 115
25    IF(L.NE.3)GOTO 26
      TYPE "LIST MONTHS TO BE FORECAST"
      ACCEPT "IF ENTIRE YEAR THEN TYPE 13 ",MONTH(3,1)
26    IF(MONTH(L,1).EQ.13)NMON(L)=12
      IF(MONTH(L,1).EQ.13)GOTO 28
      DO 27 I=2,12
        ACCEPT "NEXT MONTH? IF NONE TYPE 99 ",MONTH(L,I)
        IF(MONTH(L,I).EQ.99)GOTO 28
27      NMON(L)=NMON(L)+1                    ; NMON IS NUMBER OF MONTHS
C
28    IF(L.EQ.1)GOTO 34                      ; CHECK FOR GROUPS THE SAME
      NMN=NMON(L)
      DO 29 I=1,NMN
        IF(MONTH(1,I).NE.MONTH(L,I))GOTO 30
29      CONTINUE
      ACCEPT "SAME YEAR AS 1ST GROUP? 1=YES 2=NO ",ISAME(L)
30    IF(L.EQ.2)GOTO 34
      DO 31 I=1,NMN
        IF(MONTH(2,I).NE.MONTH(3,I))GOTO 34
31      CONTINUE
      ACCEPT "SAME YEAR AS 2ND GROUP? 1=YES 2=NO ",ISAME(3)
C
C
34    IF(MONTH(L,1).EQ.13)GOTO 38
      IF(NMON(L).EQ.1)GOTO 38
      K3=NMON(L)
      DO 36 I=2,K3
36      IF(MONTH(L,I).LT.MONTH(L,(I-1)))K1(L)=I  ; K1 IS FIRST MONTH OF YEAR
C
38    CALL GCHN(IC,IER)
      IF(L.NE.1)GOTO 40
      CALL OPENR(IC,"MONTHDAT1.DT",0,IER)
40    IF(L.NE.2)GOTO 42
      CALL OPENR(IC,"MONTHDAT2.DT",0,IER)
42    IF(L.NE.3)GOTO 44
      CALL OPENR(IC,"MONTHDAT3.DT",0,IER)
C
C
C
44    IF(MONTH(L,1).EQ.13)MONTH(L,1)=1
      DO 100 J=1,150
        IF(J.EQ.1) ISUM=0
        CALL RDL(IC,IBUF,N,IER)
        IF(IER.NE.1)GOTO 110
        CALL UNPACK(IBUF,N,IU)
        IF(J.NE.1)GOTO 48                  ; GET STARTING YEAR NEXT LINE
        ISTYR=(IU(1)-48)*1000+(IU(2)-48)*100+(IU(3)-48)*10+IU(4)-48
48      K3=NMON(L)
        K2=K1(L)
        DO 50 I=K2,K3                      ; SUM MONTHS
          ISUM=(IU(2+MONTH(L,I)*4)-48)*100+ISUM ; CONVERT TO DECIMAL
          ISUM=(IU(3+MONTH(L,I)*4)-48)*10+ISUM
          ISUM=IU(4+MONTH(L,I)*4)-48+ISUM
          IF(IU(1+MONTH(L,I)*4).EQ.45)ISUM=ISUM*(-1)
```

```
50          IF(IU(1+MONTH(L,I)*4).EQ.49)ISUM=ISUM+1000
            TSUM=(ISUM*1.0)/NMON(L)+0.5
            AVG(L,J)=TSUM
C
            ISUM=0
            IF(K1(L).EQ.1)GOTO 100              ; MONTHS ARE ALL IN SAME YEAR
            KX=K1(L)-1
            DO 55 I=1,KX                        ; GET SUM STARTED OVER NEW YEAR
              ISUM=(IU(2+MONTH(L,I)*4)-48)*100+ISUM
              ISUM=(IU(3+MONTH(L,I)*4)-48)*10+ISUM
55            ISUM=IU(4+MONTH(L,I)*4)-48+ISUM
100       CONTINUE
110       CONTINUE
          CALL KLOSE
115       CONTINUE
C
C
          J=J-1                                 ; J IS TOTAL YEARS OF RECORD
          PD(1)=J
          IF(NPDS.EQ.1)GOTO 130
          PD(1)=MV(1)-ISTYR                     ; GET PERIODS OF RECORD
          MV(NPDS)=ISTYR+J
          DO 120 I=2,NPDS
120         PD(I)=MV(I)-MV(I-1)
130       DO 200 L=1,3
          JYR=0                                 ; JYR IS START OF PERIOD
               IF(L.EQ.2.AND.MONTH(2,1).EQ.99)GOTO 200
          DO 190 I=1,NPDS
            IF(PD(I).LT.NAV)GOTO 170
C
C
C
          KST=JYR+MIDAV
          KEND=JYR+PD(I)-(NAV-MIDAV)
          DO 150 K=KST,KEND                 ; CYCLE THROUGH MIDDLE OF PERIOD
            RAV(L,K)=0
            DO 140 K2=1,NAV
140         RAV(L,K)=RAV(L,K)+AVG(L,(K-MIDAV+K2))   ; SUM AVERAGES
150         RAV(L,K)=RAV(L,K)/NAV               ; RAV IS RUNNING AVERAGE
          IF(NAV.EQ.1)GOTO 185
C
          KEND=MIDAV-1
          DO 154 K=1,KEND
154         RAV(L,(JYR+K))=RAV(L,(JYR+MIDAV))   ; FILL IN BEGINNING OF PERIOD
C
          KEND=MIDAV
          DO 156 K=1,KEND
156         RAV(L,(PD(I)-K+1+JYR))=RAV(L,(JYR+PD(I)-MIDAV+1)) ; FILL IN END
          GOTO 185
C
C
C.....PERIOD LESS THAN AVERAGING PERIOD
170       RAVG=0
          KST=JYR+1
          KEND=JYR+PD(I)
          DO 175 K=KST,KEND
175         RAVG=RAVG+AVG(L,K)
          RAVG=RAVG/PD(I)
          DO 180 K=KST,KEND
180         RAV(L,K)=RAVG
```

```
185    JYR=JYR+PD(I)
190    CONTINUE
200    CONTINUE
C
C
C
C
C
C.............COMPUTE STANDARD DEVIATION FROM RUNNING AVERAGE
       DO 230 L=1,3
         IF(L.EQ.2.AND.MONTH(2,1).EQ.99)GOTO 230
         SD(L)=0
         DO 210 I=2,J
210        SD(L)=SD(L)+(1.0*RAV(L,I)-1.0*AVG(L,I))**2 ; VARIANCE
         SD(L)=SQRT(SD(L)/(J-1))                    ; STANDARD DEVIATION
         DO 220 I=2,J          ; COMPUTE CATEGORY 1=MUCH BELOW
           DIFF=1.0*RAV(L,I)-1.0*AVG(L,I)
           ICAT(L,I)=1
           IF(DIFF.LT.(1.15)*SD(L))ICAT(L,I)=2
           IF(DIFF.LT.(.67)*SD(L))ICAT(L,I)=3
           IF(DIFF.LE.(-.67)*SD(L))ICAT(L,I)=4
220        IF(DIFF.LE.(-1.15)*SD(L))ICAT(L,I)=5
230        CONTINUE
C
C
       J2=0
       IF(MONTH(1,1).LT.MONTH(3,1).AND.K1(3).EQ.1.AND.
      2MONTH(1,1).LT.MONTH(2,1).AND.K1(2).EQ.1)GOTO 239
       IF(ISAME(2).NE.2.AND.ISAME(3).NE.2)GOTO 239
       J2=1
       KEND=J-J2
       DO 237 L=2,3
         IF(L.EQ.2.AND.ISAME(2).EQ.2)GOTO 237
         DO 235 I=2,KEND
           AVG(L,I)=AVG(L,(I+1))              ; PUT IN PROPER YEAR
           RAV(L,I)=RAV(L,(I+1))
235        ICAT(L,I)=ICAT(L,(I+1))
237      CONTINUE
C
C.........PUT INTO (1ST,FCST) MATRIX
239    KEND=J-J2
       DO 250 I=1,5
         DO 245 L=1,5
           MAT2(I,L)=0
           DO 240 I2=1,5
240          MAT3(I,L,I2)=0
245        CONTINUE
250      CONTINUE
C
C
C.........PRINT SECTION
       CALL DFILW("AVGPRINT.DT")
       CALL CRAND("AVGPRINT.DT")
       CALL GCHN(JC,IER)
       CALL OPEN(JC,"AVGPRINT.DT",0,IER)
C
C
       IF(MONTH(2,1).NE.99)GOTO 300
       DO 260 I=2,KEND
260      MAT2(ICAT(1,I),ICAT(3,I))=MAT2(ICAT(1,I),ICAT(3,I))+1
```

```
C
C
      WRITE(JC,900)
900   FORMAT("                FORECAST")
      WRITE(JC,901)
901   FORMAT("             MB   B  NM   A  MA")
      WRITE(JC,902) (MAT2(1,I),I=1,5)
902   FORMAT(" 1ST    MB",5I4)
      WRITE(JC,903) (MAT2(2,I),I=1,5)
903   FORMAT(" PER     B",5I4)
C
      WRITE(JC,904) (MAT2(3,I),I=1,5)
904   FORMAT("       NORM",5I4)
      WRITE(JC,905) (MAT2(4,I),I=1,5)
905   FORMAT("          A",5I4)
      WRITE(JC,906) (MAT2(5,I),I=1,5)
906   FORMAT("         MA",5I4)
      WRITE(JC,907)
C
C
907   FORMAT("  ")                          ; INTERVALS
      SD1=.067*SD(1)
      SD3=.067*SD(3)
      WRITE(JC,908) SD1,SD3
908   FORMAT("       ABOVE INTERVAL  OBS:",F4.1," FCST:",F4.1)
      SD1=.115*SD(1)
      SD3=.115*SD(3)
      WRITE(JC,909) SD1,SD3
909   FORMAT(" MUCH ABOVE INTERVAL  OBS:",F4.1," FCST:",F4.1)
C
C
      WRITE(JC,907)
      WRITE(JC,910)                              ; LIST YEARS
910   FORMAT("  YEAR  AVG  RAVG  AVG RAVG")
      KEND=J-J2
      DO 270 I=2,KEND
        PAVG1=AVG(1,I)*0.1
        PRAV1=RAV(1,I)*0.1
        PAVG3=AVG(3,I)*0.1
        PRAV3=RAV(3,I)*0.1
        IYR=ISTYR+I-1
270     WRITE(JC,911) IYR,PAVG1,PRAV1,PAVG3,PRAV3
911   FORMAT(I6,4F5.1)
      GOTO 450
C
C
C.....3 CATEGORY PRINT
300   DO 310 I=2,KEND
310     MAT3(ICAT(1,I),ICAT(2,I),ICAT(3,I))=MAT3(ICAT(1,I),ICAT(2,I),
     2ICAT(3,I))+1
      PD(1)="MB"
      PD(2)=" B"
      PD(3)=" N"
      PD(4)=" A"
      PD(5)="MA"
      WRITE(JC,907)
      WRITE(JC,912)
912   FORMAT("               FORECAST")           ;TITLES
      WRITE(JC,913)
913   FORMAT(" 1ST 2ND   MB  B  N  A MA")
```

```
      DO 330 I1=1,5
        DO 320 I2=1,5
320       WRITE(JC,914) PD(I1),PD(I2),(MAT3(I1,I2,I),I=1,5) ; DATA
914   FORMAT(2X,S2,2X,S2,2X,5I3)
330     CONTINUE
C
C
C
C
      WRITE(JC,907)                                ; INTERVALS
      DO 340 I=1,3
        PA(I)=SD(I)*.067
340     PS(I)=SD(I)*.115
      WRITE(JC,915) (PA(I),I=1,3)
915   FORMAT("      ABOVE INTERVAL  1ST:",F4.1," 2ND",F4.1," FCST",F4.1)
      WRITE(JC,916) (PS(I),I=1,3)
916   FORMAT(" MUCH ABOVE INTERVAL  1ST:",F4.1," 2ND",F4.1," FCST",F4.1)
C
C
      WRITE(JC,907)
      WRITE(JC,917)                                ; LIST YEARS
917   FORMAT("  YEAR  AVG RAVG  AVG RAVG  AVG RAVG")
      DO 360 I=2,KEND
        DO 350 L=1,3
          PA(L)=AVG(L,I)*0.1
350       PS(L)=RAV(L,I)*0.1
        IYR=ISTYR+I-1
360     WRITE(JC,918) IYR, (PA(I1),PS(I1),I1=1,3)
918   FORMAT(I6,6F5.1)
C

C
C
450   CALL KLOSE
      TYPE "OUTPUT IN AVGPRINT.DT"
      STOP
      END
```

Eastern Region Computer Programs and Problems (Continued)

19 Verification of Asynchronous Transmissions. Lawrence Cedrone, March 1984. (PB84 189885)

20 AFOS Hurricane Plotter. Charles Little, May 1984. (PB84 199629)

21 WARN - A Warning Formatter. Gerald G. Rigdon, June 1984. (PB84 204551)

22 Plotting TDL Coastal Wind Forecasts. Paula Severe, June 1984 (Revised) (PB84 220789)

23 Severe Weather Statistics STADTS Decoder (SWX) and Plotter (SWY). Hugh M. Stone, June 1984. (PB84 213693)

24 WXR. Harold Opitz, August 1984. (PB84 23722) (Revised August 1985, PB84 100815/AS)

25 FTASUM: Aviation Forecast Summaries. Matthew Peroutka, August 1984. (PB85 112977)

26 SAOSUM: A Short Summary of Observations. Matthew Peroutka, October 1984. (PB85 120384)

27 TRAJ - Single Station Trajectory Plot. Tom Niziol, December 1984. (PB85 135002)

28 VIDTEX. Gerald G. Rigdon, February 1985. (PB85 175669/AS)

29 Isentropic Plotter. Charles D. Little, February 1985. (PB85 175651/AS)

30 CERR: An Aviation Verification Program. M. Peroutka, April 1985. (PB85 204824/AS)

31 Correlation and Regression Equation - REGRS. Hugh M. Stone, May 1985. (PB85 213353/AS)

32 Scatter Diagram and Histogram Program - SCATR. Hugh M. Stone, May 1985. (PB85 213346/AS)

33 TIMCHEK. Gerald G. Rigdon, June 1985. (PB85-221257/AS)

34 A MOS Temperature - PoP Forecast Plot. William C. Randel, October 1985. (PB86 120029/AS)

35 ROTODRAW. Thomas Niziol, November 1985 (PB86 131828/AS)

36 LAWEB: Data Processing for the Great Lakes. William C. Randel and Matthew R. Peroutka, March 1986. (PB86 176658/AS)

37 Convective Parameters & Hodograph Program - Convect. Hugh M. Stone, April 1986. (PB86-197225/AS)

38 DWXR - SHEF Product Compression Program. Harold H. Opitz, September 1986.

39 CRASHQ: Listing Products Being Transmitted At the Time of a Crash. William C. Randel, January 1987

NOAA SCIENTIFIC AND TECHNICAL PUBLICATIONS

The National Oceanic and Atmospheric Administration was established as part of the Department of Commerce on October 3, 1970. The mission responsibilities of NOAA are to assess the socioeconomic impact of natural and technological changes in the environment and to monitor and predict the state of the solid Earth, the oceans and their living resources, the atmosphere, and the space environment of the Earth.

The major components of NOAA regularly produce various types of scientific and technical information in the following kinds of publications:

PROFESSIONAL PAPERS—Important definitive research results, major techniques, and special investigations.

CONTRACT AND GRANT REPORTS—Reports prepared by contractors or grantees under NOAA sponsorship.

ATLAS—Presentation of analyzed data generally in the form of maps showing distribution of rainfall, chemical and physical conditions of oceans and atmosphere, distribution of fishes and marine mammals, ionospheric conditions, etc.

TECHNICAL SERVICE PUBLICATIONS—Reports containing data, observations, instructions, etc. A partial listing includes data serials; prediction and outlook periodicals; technical manuals, training papers, planning reports, and information serials; and miscellaneous technical publications.

TECHNICAL REPORTS—Journal quality with extensive details, mathematical developments, or data listings.

TECHNICAL MEMORANDUMS—Reports of preliminary, partial, or negative research or technology results, interim instructions, and the like.

Information on availability of NOAA publications can be obtained from:

NATIONAL TECHNICAL INFORMATION SERVICE
U. S. DEPARTMENT OF COMMERCE
5285 PORT ROYAL ROAD
SPRINGFIELD, VA 22161